Behavior Research Methods & Instrumentation
1978, Vol. 10 (3), 426-428

PSYCHIC: A BASIC game to test ESP as d'

DONALD J. POLZELLA and ALLEN S. GOUSE
University of Dayton, Dayton, Ohio 45469

In the language of communication theory, extrasensory perception (ESP) can be defined as receiving "nonsensory" information over a discrete communication channel containing noise, where noise represents whatever it is that causes errors. This conception of ESP may be modeled by the theory of signal detection (TSD) (Green & Swets, 1966), a psychophysical paradigm that yields a measure of an individual's sensitivity to the presence of a discrete stimulus (signal). In the typical TSD experiment, a subject is asked to decide on each of many trials whether or not a near-threshold stimulus (e.g., a tone, a light, etc.) was presented within a prescribed interval. It is assumed that the subject's environment produces some background stimulation, or noise, the effect of which is variable and Gaussian. Further, it is assumed that the effect of a signal added to the noise is also variable and Gaussian. Thus, a subject decides on each trial whether he/she received only noise (N) or signal plus noise (SN).

The TSD experiment yields two statistics: d', a measure of sensitivity that is equal to the distance between the means of the SN and N distributions in standard normal deviates, and β, a measure of response threshold that is equal to the ratio of the ordinates of the SN to N distributions at the criterion (i.e., "yes-no" threshold). These statistics are empirically derived from the observed proportion of hits (the subject says "yes" on SN trials), misses (the subject says "no" on SN trials), correct rejections (the subject says "no" on N trials), and false alarms (the subject says "yes" on N trials).

The authors wish to thank L. Castelli and D. Marchaj for their help in programming earlier versions of PSYCHIC. Allen S. Grouse is now at Butler Hospital, Providence, Rhode Island 02906. Correspondence should be sent to D. J. Polzella, Department of Psychology, University of Dayton, Dayton, Ohio 45469.

TSD has been applied to a number of areas besides sensory psychophysics, including learning, memory, attention, and clinical psychology (Pastore & Scheirer, 1974). PSYCHIC represents the application of TSD to ESP research. Instead of asking subjects to decide whether or not a signal was presented, PSYCHIC asks subjects to predict if a signal (i.e., the ringing of the bell on the computer terminal) will be presented.

Program Input. The PSYCHIC program is shown in Table 1. The first input (Line 40) is the number of trials in the experiment. PSYCHIC accommodates any number of trials. The next four inputs (Lines 60, 80, 110, 140) are the payoff matrix values (positive or negative integers) applied to each trial outcome (hit, miss, correct rejection, and false alarm, respectively). By assigning various payoffs, the investigator can presumably alter β without affecting d'. The remaining inputs (Line 270) are the subject's responses, "Y" (yes) and "N" (no).

Program Output. A 10-trial sample run of PSYCHIC is shown in Table 2. PSYCHIC initiates each trial by printing a "?." The subject responds by typing "Y" (the bell will ring) or "N" (the bell will not ring). PSYCHIC generates a random integer between 0 and 1 (Line 340) and prints either "N" and no bell (Line 370) or "Y" accompanied by a bell (Line 490). Following the last trial, PSYCHIC prints the frequencies of the various outcomes (Lines 610-780) and computes points earned based on the payoff matrix values.

Restrictions. PSYCHIC generates N and SN trials with equal probability, but this can be easily modified. While PSYCHIC will accommodate any number of trials, we recommend at least 250, to insure valid statistical analysis.

Table 1
PSYCHIC

```
LIST
   10  REM INTRODUCTION FOR THE EXPERIMENTER
   20  PRINT "HOW MANY GUESS TRIALS WOULD YOU CARE TO"
   30  PRINT "HAVE THIS CANDIDATE DO?"
   40  INPUT C
   50  PRINT "PLEASE ENTER THE PAYOFF MATRIX VALUE FOR HITS."
   60  INPUT C1
   70  PRINT "PLEASE ENTER THE PAYOFF MATRIX VALUE FOR MISSES."
   80  INPUT C2
   90  PRINT "PLEASE ENTER THE PAYOFF MATRIX VALUE FOR CORRECT"
  100  PRINT "REJECTIONS."
  110  INPUT C3
  120  PRINT "PLEASE ENTER THE PAYOFF MATRIX VALUE FOR FALSE"
  130  PRINT "ALARMS."
  140  INPUT C4
  150  PRINT
  160  PRINT "THANK YOU. WE WILL NOW PROCEED WITH THE TESTING."
```

Table 1 Continued

```
170             STOP
180             PRINT
190             PRINT
200             READ R$, S$
210             LET K=1
220             LET L=1
230             LET M=1
240             LET N=1
250             FOR D=1 TO C
260             REM THE PERSON IS ASKED TO GUESS
270             INPUT X$
280             IF X$="Y" THEN 340
290             IF X$=S$ THEN 340
300             PRINT "YOU HAVE MADE AN ERROR IN RESPONDING. TRY AGAIN."
310             PRINT
320             GOTO 270
330             REM THE COMPUTER GENERATES A 'Y' OR AN 'N'
340             LET Z=INT(2*RND(-1)+0)
350             IF Z=1 THEN 490
360             REM PART CONCERNED WITH COMPUTER RESPONSE 'N'
370             PRINT S$
380             PRINT
390             IF X$=S$ THEN 450
400             REM FALSE ALARMS COUNTED
410             LET K1=K
420             LET K=K+1
430             GOTO 590
440             REM CORRECT REJECTIONS COUNTED
450             L1=L
460             LET L=L+1
470             GOTO 590
480             REM PART CONCERNED WITH COMPUTER RESPONSE 'Y'
490             PRINT R$
500             PRINT
510             IF X$="Y" THEN 570
520             REM MISSES COUNTED
530             LET M1=M
540             LET M=M+1
550             GOTO 590
560             REM HITS COUNTED
570             LET N1=N
580             LET N=N+1
590             NEXT D
600             REM THE PRODUCTS AND SUM OF POINTS IS COMPUTED
610             LET P1=C1*N1
620             LET P2=C2*M1
630             LET P3=C3*L1
640             LET P4=C4*K1
650             LET S=P1+P2+P3+P4
660             DATA Y,N
670             PRINT
680             PRINT "EVENT";TAB(18);"FREQUENCY";TAB(36);"PAYOFF";
690             PRINT "#VALUES";TAB(54);"POINTS EARNED"
700             PRINT "*****";TAB(18);"*********";
710             PRINT TAB(36);"****** ******";TAB(54)"****** ******"
720             PRINT
730             PRINT "HITS";TAB(18);N1;TAB(36);C1;TAB(54);P1
740             PRINT "MISSES";TAB(18);M1;TAB(36);C2;TAB(54);P2
750             PRINT "COR.REJECTIONS";TAB(18);L1;TAB(36);C3;TAB(54);P3
760             PRINT "FALSE ALARMS";TAB(18);K1;TAB(36);C4;TAB(54);P4
770             PRINT
780             PRINT "THE SUM OF POINTS EARNED IS:              ";S
790             END
*
```

Table 2
A Sample Run of Psychic

```
HOW MANY GUESS TRIALS WOULD YOU CARE TO
HAVE THIS CANDIDATE DO?
?10
PLEASE ENTER THE PAYOFF MATRIX VALUE FOR HITS.
?1
PLEASE ENTER THE PAYOFF MATRIX VALUE FOR MISSES.
?-1
PLEASE ENTER THE PAYOFF MATRIX VALUE FOR CORRECT
REJECTIONS.
?1
PLEASE ENTER THE PAYOFF MATRIX VALUE FOR FALSE ALARMS.
?-1
THANK YOU. WE WILL NOW PROCEED WITH THE TESTING.
STOP AT 00170, END OR CONT--CONT
?N
Y
?Y
N
?N
N
?Y
Y
?Y
Y
?N
Y
?X
YOU HAVE MADE AN ERROR IN RESPONDING. TRY AGAIN.
?N
Y
?N
N
?N
N
?Y
Y
```

EVENT	FREQUENCY	PAYOFF VALUES	POINTS EARNED
******	**********	**********	**********
HITS	3	1	3
MISSES	3	-1	-3
COR.REJECTIONS	3	1	3
FALSE ALARMS	1	-1	-1
THE SUM OF POINTS EARNED IS:		2	

Software and Hardware. PSYCHIC was written in BASIC and developed on a standard Teletype coupled to a Univac 7 computer.

Data. In order to test for ESP, the raw data can be converted to number correct and analyzed using traditional means of assessing binomial variability. Alternatively, the data can be converted to d' and the hypothesis $d' > 0$ can be tested using a statistic derived by Marascuilo (1970). A subject's isosensitivity curve (receiver operating characteristic) can be generated following multiple runs of PSYCHIC at various payoff contingencies.

REFERENCES

GREEN, D. M., & SWETS, J. A. *Signal detection theory and psychophysics*. New York: Wiley, 1966.

MARASCUILO, L. A. Extensions of the significance test for one-parameter signal detection hypotheses. *Psychometrika*, 1970, **35**, 237-243.

PASTORE, R. E., & SCHEIRER, C. J. Signal detection theory: Considerations for general application. *Psychological Bulletin*, 1974, **81**, 945-958.

(Accepted for publication March 3, 1978.)